Infinite Potential Wall

The energy eigenfunctions and eigenvalues of the infinite potetial wall:

```
ψ[x_, n_] = Sqrt[2/L] Sin[n π/L x]; En[n_] = ħ^2 π^2/(2 m L^2) n^2;
nrule = {L → 1, m → 1, ħ → 1, p → 10 π}; (*Numerical values chosen for plots*)
Ncut = 40; (* We truncate the Hilbert space dimension to this value *)
$Assumptions = n ∈ Integers && l ∈ Integers && n > 0 && m > 0 && L > 0 &&
   ħ > 0 && l > 0 && p ∈ Reals && x ∈ Reals && y ∈ Reals && t ∈ Reals;
```

Problem 4:

Consider the particle in the initial state $\phi(x,0) = (c_1\,\psi_1(x) + c_3\,\psi_3(x) + c_5\,\psi_5(x))\,e^{i\,px/\hbar}$ for $c_3/c_1 = -1/3,\ c_5/c_1 = 1/5$.
Compute the state $\phi(x,t)$ and the corresponding average values of position , momentum etc.
It is similar to problem 3 but we give to the particle initial momentum <p>=p (see nrule above).
Since now the coefficients c_n are infinite in number and complex in algebraic form (you can easily try to compute them), we put numerical values from the beginning.

Solution:

Consider the expression of the wave function interms of the energy eigenfunc- tions:

$$\phi(x,0) \equiv \phi(x) = \sum_{n=1}^{\infty} c_n\,\psi_n(x),\ \ c_n = \int_0^L \psi_n^*(x)\,\phi(x)\,dx$$

$$\phi(x,t) = \sum_{n=1}^{\infty} c_n\,\psi_n(x)\,e^{-i\,E_n\,t/\hbar}$$

$$< x > (t) = \sum_{n,m=1}^{\infty} c_m^*\,c_n\,e^{-i\,(E_n-E_m)\,t/\hbar}\int_0^L \psi_m^*(x)\,x\,\psi_n(x)\,dx$$

$$< x^2 > (t) = \sum_{n,m=1}^{\infty} c_m^*\,c_n\,e^{-i\,(E_n-E_m)\,t/\hbar}\int_0^L \psi_m^*(x)\,x^2\,\psi_n(x)\,dx$$

We compute the normalization factor A and define the wave function at t=0

```
A = 1/Sqrt[∫_0^L (ψ[x, 1] - (1/3) ψ[x, 3] + (1/5) ψ[x, 5])^2 dx]
```

$$\frac{15}{\sqrt{259}}$$

```
ϕ[x_, 0] = A (ψ[x, 1] - (1/3) ψ[x, 3] + (1/5) ψ[x, 5]) e^(i p x/ħ);
```

```
Plot[{Abs[ϕ[x, 0]] /. nrule, Re[ϕ[x, 0]] /. nrule, Im[ϕ[x, 0]] /. nrule},
 {x, 0, L /. nrule}]
```



We compute the coefficients c_n:

```
Table[cn[n] = Chop[NIntegrate[ψ[x, n] ϕ[x, 0] /. nrule,
    {x, 0, L /. nrule}, MaxRecursion → 12, WorkingPrecision → 20]], {n, 1, Ncut}]
```

```
{0, -0.0091041459728254188814 ⅈ, 0, -0.045579238960973260671 ⅈ,
 -0.093205464900180006874, 0.11363140875928066422 ⅈ, 0.15534244150030001146,
 -0.28626773333153750010 ⅈ, -0.46602732450090003537, 0.55272149331647024433 ⅈ,
 0.46602732450090003473, -0.28693353224183268740 ⅈ, -0.15534244150030001181,
 0.11220042387836555209 ⅈ, 0.093205464900180009607, -0.048018106366122356205 ⅈ,
 0, -0.013070562371201964078 ⅈ, 0, -0.0067124128550598689079 ⅈ, 0,
 -0.0041509783964305317666 ⅈ, 0, -0.0028187362409620481210 ⅈ, 0,
 -0.0020293110011210501346 ⅈ, 0, -0.0015219770315143845094 ⅈ, 0,
 -0.0011770045327498171830 ⅈ, 0, -0.00093236651814486255625 ⅈ, 0,
 -0.00075310007420444014124 ⅈ, 0, -0.00061821721995897978963 ⅈ,
 0, -0.00051448513350978892371 ⅈ, 0, -0.00043322395036926308595 ⅈ}
```

```
ϕ[x_, t_] = Sum[cn[n] ψ[x, n] ⅇ^(-ⅈ En[n] t/ℏ), {n, 1, Ncut}];
```

```
Absϕ[x_, t_] = Abs[ϕ[x, t]] /. nrule;
Manipulate[
 Plot[Absϕ[x, t]^2, {x, 0, 1}, PlotRange → {{0, 1}, {0, 4.5}}], {t, 0, 4, 0.0001}]
```




```
Reϕ[x_, t_] = Re[ϕ[x, t]] /. nrule;
Imϕ[x_, t_] = Im[ϕ[x, t]] /. nrule;
Manipulate[Plot[{Reϕ[x, t], Imϕ[x, t]}, {x, 0, 1},
  PlotRange → {{0, 1}, {-2, 2.2}}, PlotStyle → {Red, Blue}], {t, 0, 4, 0.0001}]
```



Average energy $< E >$, $< E^2 >$ and uncetainty Δ E

```
avE = Sum[Abs[cn[n]]^2 En[n], {n, 1, Ncut}]; avE2 = Sum[Abs[cn[n]]^2 En[n]^2, {n, 1, Ncut}]; Δ E = Sqrt[avE2 - avE^2] ;
Print[
 "<E> = ", avE, " = ", avE / En[1], " E1",
 "\n<E^2> = ", avE2, " = ", avE2 / En[1]^2, " E1^2",
 "\nΔ E = ", Δ E, " = ", Δ E/ En[1] // Simplify, " E1"
]
```

$\langle E\rangle = \frac{506.3331793667725107\,\hbar^2}{L^2\,m} = 102.60455410165901543\ E_1$

$\langle E^2\rangle = \frac{282\,158.16249011964793\,\hbar^4}{L^4\,m^2} = 11\,586.522756551630062\ E_1{}^2$

$\Delta E = 160.5766918405514549\sqrt{\frac{\hbar^4}{L^4\,m^2}} = 32.5396409653111903\ E_1$

Position and momenta matrix elements:

$x_{nm} = \int_0^L \psi_n^*(x)\, x\, \psi_m(x)\, dx,\ x^2{}_{nm} = \int_0^L \psi_n^*(x)\, x^2\, \psi_m(x)\, dx,\ p_{nm} = \int_0^L \psi_n^*(x)\, \hat{p}\, \psi(x)\, dx,\ p^2{}_{nm} = \int_0^L \psi_n^*(x)\, \hat{p}^2\, \psi(x)\, dx$

```
xnm[n_, l_] = Integrate[ψ[x, n] x ψ[x, l], {x, 0, L}]; xnm[n_, n_] = Integrate[ψ[x, n] x ψ[x, n], {x, 0, L}];

x2nm[n_, l_] = Integrate[ψ[x, n] x^2 ψ[x, l], {x, 0, L}]; x2nm[n_, n_] = Integrate[ψ[x, n] x^2 ψ[x, n], {x, 0, L}];

pnm[n_, l_] = Integrate[ψ[x, n] (-ⅈ ℏ) ∂x ψ[x, l], {x, 0, L}] /. Cos[l π] → (-1)^l;

pnm[n_, n_] = Integrate[ψ[x, n] (-ⅈ ℏ) ∂x ψ[x, n], {x, 0, L}] ;

p2nm[n_, l_] = Integrate[ψ[x, n] (- ℏ^2) ∂xx ψ[x, l], {x, 0, L}] ;

p2nm[n_, n_] = Integrate[ψ[x, n] (- ℏ^2) ∂xx ψ[x, n], {x, 0, L}] ;
```

```
Table[pnm[i, j], {i, 1, 5}, {j, 1, 5}] // MatrixForm
```

$$\begin{pmatrix} 0 & \frac{8 i \hbar}{3 L} & 0 & \frac{16 i \hbar}{15 L} & 0 \\ -\frac{8 i \hbar}{3 L} & 0 & \frac{24 i \hbar}{5 L} & 0 & \frac{40 i \hbar}{21 L} \\ 0 & -\frac{24 i \hbar}{5 L} & 0 & \frac{48 i \hbar}{7 L} & 0 \\ -\frac{16 i \hbar}{15 L} & 0 & -\frac{48 i \hbar}{7 L} & 0 & \frac{80 i \hbar}{9 L} \\ 0 & -\frac{40 i \hbar}{21 L} & 0 & -\frac{80 i \hbar}{9 L} & 0 \end{pmatrix}$$

Here we have to be careful: Due to large arguments in trig/exp functions the numerical errors are large. In particular we find large real parts. Therefore we compute the real part from the beginning. Notice the use of the Refine[] function in order for mathematica to use the fact that t is real and compute the real part.

```
δ[i_, j_] := KroneckerDelta[i, j]; δ2[i_, j_] := (2 - δ[i, j]);
xavϕ[t_] =
  Sum[Sum[(δ2[n, l] Refine[Re[cn[n] cn[l] xnm[n, l] e^(-i (En[l]-En[n]) t/ħ)]]), {l, n, Ncut}], {n, 1, Ncut}] /. nrule // N;
x2avϕ[t_] = Sum[Sum[(δ2[n, l] Refine[Re[cn[n] cn[l] x2nm[n, l] e^(-i (En[l]-En[n]) t/ħ)]]), {l, n, Ncut}], {n, 1, Ncut}] /.
    nrule // N;
pavϕ[t_] = Sum[Sum[(δ2[n, l] Refine[Re[cn[n] cn[l] pnm[n, l] e^(-i (En[l]-En[n]) t/ħ)]]), {l, n, Ncut}], {n, 1, Ncut}] /.
    nrule // N;
p2avϕ[t_] = Sum[Sum[(δ2[n, l] Refine[Re[ cn[n] cn[l] p2nm[n, l] e^(-i (En[l]-En[n]) t/ħ)]]), {l, n, Ncut}], {n, 1, Ncut}] /.
    nrule // N;
Δx[t_] = Sqrt[x2avϕ[t] - xavϕ[t]^2]; Δp[t_] = Sqrt[p2avϕ[t] - pavϕ[t]^2];
```

```
{Plot[xavϕ[t] /. nrule, {t, 0, 1},
  PlotLabel → "<x>(t)", PlotStyle → Directive[Red, Thick]],
 Plot[√(x2avϕ[t] /. nrule), {t, 0, 1}, PlotLabel → "<x^2>^1/2(t)",
  PlotStyle → Directive[Blue, Thick]]}
```



This should be the same as the value of p in nrule:

```
pavϕ[0]
```

```
31.4158
```

```
{Plot[pavϕ[t] /. nrule, {t, 0, 1},
  PlotLabel → "<p>(t)", PlotStyle → Directive[Red, Thick]],
 Plot[√(p2avϕ[t] /. nrule), {t, 0, 1}, PlotLabel → "<p^2>^1/2(t)",
  PlotStyle → Directive[Blue, Thick]]}
```



```
{Plot[Δ x[t] /. nrule, {t, 0, 1},
  PlotLabel → "Δ x(t)", PlotStyle → Directive[Red, Thick]],
 Plot[Δ p[t] /. nrule, {t, 0, 1}, PlotLabel → "Δ p(t)",
  PlotStyle → Directive[Blue, Thick]],
 Plot[{Δ x[t] Δ p[t] /. nrule, ħ / 2 /. nrule}, {t, 0, 1}, PlotLabel → "Δ x(t) Δ p(t)",
  PlotStyle → {Directive[Magenta, Thick], Directive[Green, Thick]}]}
```